///////////////////////////////////////////////////
// Election Administration Polarization
// Diversity RD Table
///////////////////////////////////////////////////

gl path "~/Library/CloudStorage/GoogleDrive-danmckinleythompson@gmail.com/My Drive/ElecAdminPolarization/How_Partisan_Is_Local_Election_Admin_Replication"

* Bring in the presidential election analysis data
use "$path/analysis_data/rd_election_analysis_data.dta", clear
keep if office=="pres"

* Define diversity
gen diverse = share_nhwhite<0.8

// Less Diverse

rdbwselect r_oos_state_year_vs_dem rv if diverse==0, vce(cluster election_id) kernel(uniform)
gl bw = e(h_mserd)

* Spec 1, Residualized Outcome
reg r_oos_state_year_vs_dem treat rv treat_rv if abs(rv)<$bw & diverse==0, vce(clust election_id)
rdtab r_oos_state_year_vs_dem treat rv election_id, est_command("reg") new("B") tau(0.02) alpha(0.05) onesided

* Spec 2, Residualized Outcome
reg r_oos_state_year_vs_dem treat rv treat_rv if abs(rv)<$bw*2 & diverse==0, vce(clust election_id)
rdtab r_oos_state_year_vs_dem treat rv election_id, est_command("reg") old("B") tau(0.02) alpha(0.05) onesided

* Spec 3, Residualized Outcome
reg r_oos_state_year_vs_dem treat rv treat_rv if abs(rv)<$bw*0.5 & diverse==0, vce(clust election_id)
rdtab r_oos_state_year_vs_dem treat rv election_id, est_command("reg") old("B") tau(0.02) alpha(0.05) onesided

* Spec 4, Residualized Outcome
rdrobust r_oos_state_year_vs_dem rv if diverse==0, vce(cluster election_id) all
rdtab r_oos_state_year_vs_dem treat rv election_id if diverse==0, est_command("rdrobust") old("B") tau(0.02) alpha(0.05) onesided


// More Diverse

rdbwselect r_oos_state_year_vs_dem rv if diverse==1, vce(cluster election_id) kernel(uniform)
gl bw = e(h_mserd)

* Spec 1, Residualized Outcome
reg r_oos_state_year_vs_dem treat rv treat_rv if abs(rv)<$bw & diverse==1, vce(clust election_id)
rdtab r_oos_state_year_vs_dem treat rv election_id, est_command("reg") old("B") tau(0.02) alpha(0.05) onesided

* Spec 2, Residualized Outcome
reg r_oos_state_year_vs_dem treat rv treat_rv if abs(rv)<$bw*2 & diverse==1, vce(clust election_id)
rdtab r_oos_state_year_vs_dem treat rv election_id, est_command("reg") old("B") tau(0.02) alpha(0.05) onesided

* Spec 3, Residualized Outcome
reg r_oos_state_year_vs_dem treat rv treat_rv if abs(rv)<$bw*0.5 & diverse==1, vce(clust election_id)
rdtab r_oos_state_year_vs_dem treat rv election_id, est_command("reg") old("B") tau(0.02) alpha(0.05) onesided

* Spec 4, Residualized Outcome
rdrobust r_oos_state_year_vs_dem rv if diverse==1, vce(cluster election_id) all
rdtab r_oos_state_year_vs_dem treat rv election_id if diverse==1, est_command("rdrobust") old("B") tau(0.02) alpha(0.05) onesided


qui {
* Build the memo/paper table
cap log close
set linesize 255
log using "$path/output/rd_table_diversity.tex", text replace
noi di "\begin{table}[ht]"
noi di "\centering"
noi di "\caption{\textbf{Effect of Democratic Election Officials on Democratic Presidential Vote Share, More vs. Less Racially and Ethnically Diverse Counties.}"
noi di "\label{tab:rd_table_diversity}}"
noi di "\begin{tabular}{lcccc|cccc}"
noi di "\toprule \toprule"
noi di " & \multicolumn{8}{c}{Dem Pres Vote Share}\\"
noi di " & \multicolumn{4}{c}{Less Diverse} & \multicolumn{4}{c}{More Diverse}\\"
noi di " & (1) & (2) & (3) & (4) & (5) & (6) & (7) & (8) \\"
noi di "\midrule"
noi di "Dem Elec Official & " %4.3f B[1,1] " & " %4.3f B[1,2] " & " %4.3f B[1,3] " & " %4.3f B[1,4] " "
noi di 		" & " %4.3f B[1,5] " & " %4.3f B[1,6] " & " %4.3f B[1,7] " & " %4.3f B[1,8] "\\"
noi di " & (" %4.3f B[2,1] ") & (" %4.3f B[2,2] ") & (" %4.3f B[2,3] ") & (" %4.3f B[2,4] ") "
noi di 		" & (" %4.3f B[2,5] ") & (" %4.3f B[2,6] ") & (" %4.3f B[2,7] ") & (" %4.3f B[2,8] ") \\[2mm]"
noi di "\midrule"
noi di "N & " %4.0f B[3,1] " & " %4.0f B[3,2] " & " %4.0f B[3,3] " & " %4.0f B[3,4] " "
noi di 		" & " %4.0f B[3,5] " & " %4.0f B[3,6] " & " %4.0f B[3,7] " & " %4.0f B[3,8] " \\ "
noi di "Bandwidth & " %4.2f B[5,1] " & " %4.2f B[5,2] " & " %4.2f B[5,3] " & " %4.2f B[5,4] " "
noi di 		" & " %4.2f B[5,5] " & " %4.2f B[5,6] " & " %4.2f B[5,7] " & " %4.2f B[5,8] " \\ "
noi di "BW Selection & CCT & CCT*2 & CCT/2 & CCT & CCT & CCT*2 & CCT/2 & CCT \\"
noi di "Kernel & Unif & Unif & Unif & Tri & Unif & Unif & Unif & Tri \\"
noi di "\bottomrule \bottomrule"
noi di "\multicolumn{9}{p{1.02\textwidth}}{\footnotesize Robust standard errors in parentheses. "
noi di "The outcome is first regressed on a state- and year-specific lag using all counties including those for which clerk election "
noi di "results are not available. The regression discontinuity is estimated using the residuals from that regression. "
noi di "The bandwidth row reports the number of maximum clerk win margin allowed for inclusion in each specificaiton. "
noi di "CCT refers to \cite{calonico2014robust} bandwidth selection procedure. "
noi di "Unif means the specification uses a uniform kernel. Tri means the specification uses a triangular kernel. "
noi di "More diverse counties are those where the non-Hispanic White residents make up less than 80\% of the population. "
noi di "All other counties are coded as less diverse.}"
noi di "\end{tabular}"
noi di "\end{table}"
log off
}

